Harjoitus 6 -- Ratkaisut

1

Ei kommenttia.

2

Haetaan data tiedostosta.

```
SetDirectory["/home/ofys/jmattas/"]
(*SetDirectory[
 "c:/documents and settings/mattas/desktop/teaching/atk2/harjoitukset/h06"]*)
```

```
/net/nfstu/home4/ofys/jmattas
```

```
c:\documents and settings\mattas\desktop\teaching\atk2\harjoitukset\h06
```

```
Clear[t, x, y]
t = Import["h06_data.txt", "Table"];
n = Length[t];
g = ListPlot[t, ImageSize → Small]
```



Sovitetaan tähän datapisteistöön suora pienimmän neliösumman menetelmällä (ks. luennot). Lasketaan sovituksessa tarvittavat summat muuttujiin sx, sy, sxy ja sxx.

```
For[i = 1; sx = 0; sy = 0; sxy = 0; sxx = 0, i ≤ n, i++,
 x = t[[i, 1]]; y = t[[i, 2]];
 sx += x;
 sy += y;
 sxy += x * y;
 sxx += x * x;]
```

Lasketaan suoran kulmakerroin ja vakiotermi:

$$a = \frac{n\,sxy - sx\,sy}{n\,sxx - sx^2};\ b = \frac{sy\,sxx - sx\,sxy}{n\,sxx - sx^2};$$

```
s[x_] := a x + b;
```

Last[t] antaa taulukon t viimeisen lukuparin. Sen 1. alkio saadaan tietysti Last[t][[1]].

```
Show[g, Plot[s[x], {x, 0, Last[t][[1]]}], ImageSize → Small]
```



Sovitus saadaan laskettua hyödyntämällä enemmän Mathematican valmiita funktioita. Merkintä t[[All,i]] antaa t:n i:nnen pystyrivin, ts. t[[All,1]] antaa kaikkien lukuparien 1. alkiot.

```
sx = Total[t[[All, 1]]];
sy = Total[t[[All, 2]]];
sxx = Total[t[[All, 1]]^2];
sxy = Total[t[[All, 1]] * t[[All, 2]]];
```

$$s[x\_] := \frac{n\,sxy - sx\,sy}{n\,sxx - sx^2}\,x + \frac{sy\,sxx - sx\,sxy}{n\,sxx - sx^2};$$

```
Show[g, Plot[s[x], {x, 0, Last[t][[1]]}], ImageSize → Small]
```



Sovitussuoran kulmakerroin on siis komponentin *konduktanssi*, resistanssin käänteisarvo:

```
Print["Konduktanssi: ", a, " Siemens"]
Print["Resistanssi: ", a^-1, " Ohm"]
```

```
Konduktanssi: 143.177 Siemens

Resistanssi: 0.00698438 Ohm
```

Yo. sovitus ei kulje origon kautta. Origon kautta kulkevalle suoralle pitää laskea sovitusparametrit uudestaan. Minimoitavaksi tulee summa

$\sum_{i=1}^{n} (a\,x_i - y_i)^2$,

josta saadaan derivoimalla a:n suhteen ja asettamalla derivaatta nollaksi:

$a = \frac{\sum x_i\,y_i}{\sum x_i^2}$.

$$a = \frac{sxy}{sxx};$$

```
Show[g, Plot[a x, {x, 0, Last[t][[1]]}], ImageSize → Small]
```



Saadaan hieman korkeampi konduktanssi.

```
Print["Konduktanssi: ", a, " Siemens"]
Print["Resistanssi: ", a^-1, " Ohm"]
```

```
Konduktanssi: 146.007 Siemens

Resistanssi: 0.00684899 Ohm
```

Tehdään sovitus vielä Mathematican valmiilla funktiolla Fit. Se ottaa argumenteikseen datapisteistön, listan funktioista joiden lineaarikombinaationa sovitusta etsitään, ja kolmantena mitä symbolia käytetään muuttujana sovituksen tuloksessa.

```
y = Fit[t, {x}, x]
```

```
146.007 x
```

Tai ei-origon-kautta-kulkeva suora:

```
y = Fit[t, {1, x}, x]
```

```
3.86818 + 143.177 x
```

Tarkistetaan tulos:

```
Show[Plot[y, {x, 0, t[[n, 1]]}], Plot[s[x], {x, 0, t[[n, 1]]}], ImageSize → Small]
```



3

Regress-funktio antaa paljon statistiikkaa sovituksen hyvyydestä. Ainut meidän kannaltamme olennainen tilastollinen parametri lienee SE, eli Standard Error. Regress-funktion argumentit ovat samat kuin Fit-funktiolle, mutta se ei palauta suoraan sovitusfunktiota. Regress-funktio löytyy seuraavasta paketista:

```
<< LinearRegression`

Regress[t, {x}, x]
```

{ParameterTable →

	Estimate	SE	TStat	PValue
1	3.86818	6.53681	0.591753	0.561376
x	143.177	5.45683	26.2381	8.88178×10^{-16}

, RSquared → 0.97452, AdjustedRSquared → 0.973104, EstimatedVariance → 198.017,

ANOVATable →

	DF	SumOfSq	MeanSq	FRatio	PValue
Model	1	136322.	136322.	688.436	8.88178×10^{-16}
Error	18	3564.3	198.017		
Total	19	139886.			

}

Huomataan, että Regress lisää automaattisesti vakiotermin sovitukseen. Se voidaan pakottaa pois asetuksella IncludeConstant->False

```
Regress[t, {x}, x, IncludeConstant → False]
```

DesignedRegress::tsos :
Warning: the total sum of squares in the ANOVATable is uncorrected (not centered on the response mean) when there is no constant term in the model; it is designated U Total.

DesignedRegress::rsqr : Warning: the RSquared and AdjustedRSquared diagnostics are redefined when there is no constant term in the model.

{ParameterTable →

	Estimate	SE	TStat	PValue
x	146.007	2.58139	56.5614	0.

,
RSquared → 0.994096, AdjustedRSquared → 0.993785, EstimatedVariance → 191.244,

ANOVATable →

	DF	SumOfSq	MeanSq	FRatio	PValue
Model	1	611 828.	611 828.	3199.19	0.
Error	19	3633.64	191.244		
U Total	20	615 461.			

}

Jos et halua lukea virheilmoituksia, käytä Quiet-funktiota.

```
Regress[t, {x}, x, IncludeConstant → False] // Quiet
```

{ParameterTable →

	Estimate	SE	TStat	PValue
x	146.007	2.58139	56.5614	0.

,
RSquared → 0.994096, AdjustedRSquared → 0.993785, EstimatedVariance → 191.244,

ANOVATable →

	DF	SumOfSq	MeanSq	FRatio	PValue
Model	1	611 828.	611 828.	3199.19	0.
Error	19	3633.64	191.244		
U Total	20	615 461.			

}

Regress palauttaa listan sääntöjä, joiden avulla päästään käsiksi itse lukuihin:

```
p = ParameterTable /. Regress[t, {x}, x]
```

	Estimate	SE	TStat	PValue
1	3.86818	6.53681	0.591753	0.561376
x	143.177	5.45683	26.2381	8.88178×10^{-16}

Arviot kulmakertoimelle ja vakiotermille:

```
Part[p, 1, 1, 1]
Part[p, 1, 2, 1]
```

3.86818

143.177

Virhearviot:

```
Part[p, 1, 1, 2]
Part[p, 1, 2, 2]
```

6.53681

5.45683

Lasketaan taulukosta R_i:

```
Clear[r]
For[i = 1, i ≤ n, i++,
 r[i] = t[[i, 1]] / t[[i, 2]]]
```

Keskiarvo:

```
For[i = 1; R = 0, i ≤ n, i++,
 R += r[i]]
R /= n
```

```
0.00682344
```

Keskihajonta:

```
For[i = 1; Σ = 0, i ≤ n, i++,
 Σ += (R - r[i])^2]
Σ /= n; Σ = Sqrt[Σ]
```

```
0.00051198
```

Tai kätevämmin, valmiilla funktioilla:

```
ρ = t[[All, 1]] / t[[All, 2]];
ρav = Mean[ρ] (*keskiarvo*)
σ = StandardDeviation[ρ] (*keskihajonta*)
```

```
0.00682344

0.00052528
```

Valmis funktio käyttää keskihajonnan kaavassa nimittäjää n-1.

4

Haetaan toinen datapisteistö, jossa on siis riippumattomia resistanssin mittauksia suurinpiirtein samalla jännitteen ja virran arvoilla. Muodostetaan erikseen taulukot virralle ja jännitteelle:

```
t = Import["h06_data2.txt", "Table"];
va = t[[All, 1]]; ia = t[[All, 2]];
```

Resistanssin keskiarvo:

```
rav = Mean[va / ia]
```

```
145.346
```

Lasketaan jännitteen ja virran suurimmat poikkeamat keskiarvosta:

```
Print["ΔV = ", η = Max[va - Mean[va]]]
Print["ΔI = ", φ = Max[ia - Mean[ia]]]
```

```
ΔV = 2.70969

ΔI = 0.077945
```

Leikitään hieman kokonaisdifferentiaalilla:

```
Clear[v, i, r]
r[v_, i_] := v/i;
dr = Dt[r[v, i]]
```

$$-\frac{v\,\mathrm{Dt}[i]}{i^2} + \frac{\mathrm{Dt}[v]}{i}$$

Virherajan laskemiseksi tämän lausekkeen molemmista termeistä on otettava itseisarvot, ja nämä itseisarvot lasketaan yhteen. Huom. Part-funktiolla saa vaikka summan termin tai tulon tekijän (minkä tahansa lausekkeen osan).

```
adr = Abs[Part[dr, 1]] + Abs[Part[dr, 2]]
```

$$\mathrm{Abs}\left[\frac{v\,\mathrm{Dt}[i]}{i^2}\right] + \mathrm{Abs}\left[\frac{\mathrm{Dt}[v]}{i}\right]$$

Korvataan Dt[i] ja Dt[v] ΔI:llä ja ΔV:llä, ja i ja v keskiarvoilla:

```
Δr = adr /. {i → Mean[ia], v → Mean[va], Dt[i] → φ, Dt[v] → η}
```

```
11.3431
```

Sovelletaan 15 yksikön sääntöä: Virheeksi saadaan 12 Ω, ja keskiarvo pyöristetään samaan tarkkuuteen:

```
Print["Resistanssi: R = ", Round[rav, 1], " Ω ± ", Round[Δr, 1], " Ω"]
```

```
Resistanssi: R = 145 Ω ± 11 Ω
```